Maple-assisted proof of formula for A296684

Robert Israel

18 December 2017

There are $2^8 = 256$ possible configurations for a 2×4 sub-array. Consider the 256×256 transition matrix T such that $T_{ij} = 1$ if the bottom two rows of a 3×4 sub-array could be in configuration i while the top two rows are in configuration j (i.e. the middle row is compatible with both i and j, and each 1 in that row is king-move adjacent to 1,2 or 3 1's), and 0 otherwise. The following Maple code computes it. I'm encoding a configuration

$$\begin{bmatrix} b_1 & b_2 & b_3 & b_4 \\ b_5 & b_6 & b_7 & b_8 \end{bmatrix}$$

as $b + 1$ where $b_1 b_2 b_3 b_4 b_5 b_6 b_7 b_8$ is the binary representation of b. The $+1$ is needed because matrix indices start at 1 rather than 0.

```
> q:= proc(a,b) local r,s,t,M,i;
     s:= floor((a-1)/16);
     if s <> (b-1) mod 16 then return 0 fi;
     s:= convert(s+16,base,2);
     r:= convert(16+floor((b-1)/16),base,2);
     t:= convert(16+ ((a-1) mod 16),base,2);
     M:= Vector(4);
     for i from 1 to 3 do if s[i] = 1 and s[i+1] = 1 then M[i]:= M
  [i]+1; M[i+1]:= 1 fi od;
     for i from 1 to 4 do if s[i]=1 then
       M[i]:= M[i]+r[i]+t[i];
       if i > 1 then M[i] := M[i]+r[i-1]+t[i-1] fi;
       if i < 4 then M[i] := M[i]+r[i+1]+t[i+1] fi;
       if M[i] =0 or M[i] > 3 then return 0 fi;
     fi od;
     1
  end proc:
  T:= Matrix(256,256, q);
```

$$T := \begin{bmatrix} \textit{256 x 256 Matrix} \\ \textit{Data Type: anything} \\ \textit{Storage: rectangular} \\ \textit{Order: Fortran\_order} \end{bmatrix} \quad \textbf{(1)}$$

Thus $a(n) = u\, T^n\, v$ where u and v are row and column vectors respectively with $u_i = 1$ for i corresponding to configurations with bottom row $(0, 0, 0, 0)$, 0 otherwise, and $v_i = 1$ for i corresponding to configurations with top row $(0, 0, 0, 0)$, 0 otherwise. The following Maple code produces these vectors.

```
> u:= Vector[row](256):
  v:= Vector(256):
```

```
for i from 0 to 15 do u[16*i+1]:= 1; v[i+1]:= 1;
od:
```

To check, here are the first few entries of our sequence.

```
> seq(u . T^n . v, n = 1 .. 10);
```

7, 145, 1162, 11478, 121477, 1210458, 12227803, 124103052, 1254382781, 12689916581 **(2)**

Now here is the minimal polynomial *P* of *T*, as computed by Maple.

```
> P:= unapply(LinearAlgebra:-MinimalPolynomial(T, t), t);
```

$$P := t \to t^{49} - 8\, t^{48} + 7\, t^{47} - 295\, t^{46} + 320\, t^{45} - 3269\, t^{44} + 8964\, t^{43} - 16461\, t^{42} + 69869\, t^{41} - 67492\, t^{40} + 187776\, t^{39} - 273575\, t^{38} - 40876\, t^{37} - 331513\, t^{36} - 788270\, t^{35} + 1690709\, t^{34} - 92294\, t^{33} + 3859480\, t^{32} + 1224374\, t^{31} - 6059952\, t^{30} + 1256215\, t^{29} - 11628627\, t^{28} + 3226624\, t^{27} + 22405233\, t^{26} - 11411148\, t^{25} - 1547053\, t^{24} + 2605689\, t^{23} - 13082896\, t^{22} + 9636974\, t^{21} + 4677683\, t^{20} - 5467348\, t^{19} - 529359\, t^{18} - 682933\, t^{17} + 1451181\, t^{16} + 736402\, t^{15} + 70615\, t^{14} - 176396\, t^{13} - 456176\, t^{12} + 22141\, t^{11} + 50804\, t^{10} + 29512\, t^{9} + 28076\, t^{8} - 7028\, t^{7} - 5528\, t^{6} - 356\, t^{5} + 128\, t^{4} + 16\, t^{3} \quad \mathbf{(3)}$$

This turns out to have degree 49. Thus we will have $0 = u\, P(T)\, T^n\, v = \sum_{i=0}^{49} p_i\, a(i+n)$ where p_i is the coefficient of t^i in $P(t)$. That corresponds to a homogeneous linear recurrence of order 49, which would hold true for any *u* and *v*. It seems that with our particular *u* and *v* we have a recurrence of order only 26, corresponding to a factor of *P*.

```
> factor(P(t));
```

$$t^3\, (t^{20} - t^{19} + 15\, t^{18} - 24\, t^{17} + 55\, t^{16} - 119\, t^{15} + 79\, t^{14} - 176\, t^{13} + 58\, t^{12} + 445\, t^{11} - 295\, t^{10} + 959\, t^{9} - 420\, t^{8} - 1455\, t^{7} + 655\, t^{6} + 435\, t^{5} + 171\, t^{4} + 124\, t^{3} - 116\, t^{2} - 42\, t - 4)\, (t^{26} - 7\, t^{25} - 15\, t^{24} - 181\, t^{23} + 141\, t^{22} - 269\, t^{21} + 2149\, t^{20} + 2006\, t^{19} + 3785\, t^{18} + 973\, t^{17} - 8803\, t^{16} - 516\, t^{15} + 2504\, t^{14} - 3529\, t^{13} + 5347\, t^{12} - 1719\, t^{11} - 3063\, t^{10} + 2485\, t^{9} + 404\, t^{8} + 697\, t^{7} - 572\, t^{6} - 47\, t^{5} - 143\, t^{4} - 82\, t^{3} + 100\, t^{2} + 10\, t - 4) \quad \mathbf{(4)}$$

```
> Q:= unapply(t^26-7*t^25-15*t^24-181*t^23+141*t^22-269*t^21+2149*
  t^20+2006*t^19+3785*t^18+973*t^17-8803*t^16-516*t^15+2504*t^14
  -3529*t^13+5347*t^12-1719*t^11-3063*t^10+2485*t^9+404*t^8+697*t^7
  -572*t^6-47*t^5-143*t^4-82*t^3+100*t^2+10*t-4, t);
```

$$Q := t \to -4 + 10\, t + t^{26} - 7\, t^{25} - 15\, t^{24} - 181\, t^{23} + 141\, t^{22} - 269\, t^{21} + 2149\, t^{20} + 2006\, t^{19} + 3785\, t^{18} + 973\, t^{17} - 8803\, t^{16} - 516\, t^{15} + 2504\, t^{14} - 3529\, t^{13} + 5347\, t^{12} - 1719\, t^{11} - 3063\, t^{10} + 2485\, t^{9} + 404\, t^{8} + 697\, t^{7} - 572\, t^{6} - 47\, t^{5} - 143\, t^{4} - 82\, t^{3} + 100\, t^{2} \quad \mathbf{(5)}$$

The complementary factor $R(t) = \frac{P(t)}{Q(t)}$ has degree 23.

```
> R:= unapply(normal(P(t)/Q(t)), t);
```

$$R := t \to (t^{20} - t^{19} + 15\, t^{18} - 24\, t^{17} + 55\, t^{16} - 119\, t^{15} + 79\, t^{14} - 176\, t^{13} + 58\, t^{12} + 445\, t^{11} - 295\, t^{10} + 959\, t^{9} - 420\, t^{8} - 1455\, t^{7} + 655\, t^{6} + 435\, t^{5} + 171\, t^{4} + 124\, t^{3} - 116\, t^{2} - 42\, t - 4)\, t^3 \quad \mathbf{(6)}$$

Now we want to show that $b(n) = u\, Q(T)\, T^n\, v = 0$ for all *n*. This will certainly satisfy the order-23 recurrence

$$\sum_{i=0}^{23} r_i\, b(i+n) = \sum_{i=0}^{23} r_i u\, Q(T)\, T^{n+i} v = u\, Q(T)\, R(T)\, T^n\, v = u\, P(T)\, T^n\, v = 0$$

where r_i are the coefficients of $R(t)$. To show all $b(n) = 0$ it suffices to show $b(0) = \ldots = b(22) = 0$. This would take some time to do naively, so it's worthwhile to do some pre-calculation.

```
> w:= u . Q(T):
  V[0]:= v:
  for i from 1 to 22 do V[i]:= T . V[i-1] od:
  seq(w . V[i],i=0..22);
```

0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0 **(7)**